
LITEcomp – aplikacje
Cyfrowy potencjometr audio 
sterowany pilotem RC5

st7.ep.com.pl

LITEcomp
LITEcomp jest prostym komputerkiem wy-
konanym na mikrokontrolerze ST7FLITE19. 

LITEcomp jest w ramach promocji dodawany 
bezpłatnie do książki „Mikrokontrolery ST7LITE 
w praktyce” (autor Jacek Bogusz). Książka jest 

dostępna w sklep.avt.pl (numer katalogowy 
KS–260905).

Schemat urządzenia przedsta-
wiono na rys. 1. Najważniejszym 
elementem prezentowanego pro-
jektu jest układ DS1807 produko-
wany przez firmę Maxim–Dallas. 
Układ ten jest podwójnym, 65–
pozycyjnym, logarytmicznym poten-
cjometrem cyfrowym sterowanym 
za pomocą interfejsu I2C. Oprócz 
układu DS1807 zastosowano po-
czwórny wzmacniacz operacyjny 
TL074 pracujący w układzie wtór-
nika napięciowego. Diody D1 i D2 
pełnią funkcję bramki AND i za-
stosowane są w celu umożliwienia 
generowania przerwania przez oby-
dwa przyciski modułu LITEcomp 
jednocześnie. Jest to podyktowane 
faktem, iż przyciski te są podłą-
czone do tej samej grupy wypro-
wadzeń mogących generować prze-
rwanie EI0 i w danej chwili tylko 
jeden z nich może być źródłem 
przerwania. Zastosowanie bramki 
AND i podłączenie jej wyjścia do 
wyprowadzenia z innej grupy prze-
rwań zewnętrznych spowoduje, iż 
obydwa przyciski będą mogły ge-

Prezentowany w artykule, 
sterowany pilotem RC5, regulator 

głośności jest kompilacją 
procedur przedstawianych 

w trzech poprzednich 
asemblerowych przykładach 

aplikacji modułu LITEcomp. 
Z tego powodu artykuł będzie 
nieco krótszy niż poprzednie, 

gdyż większość wykorzystywanych 
procedur została już omówiona 
w poprzednich odcinkach cyklu. 

nerować przerwanie. Dzielnik na-
pięcia na rezystorach R1 i R2 wy-
twarza pozorną masę analogową 
w połowie napięcia zasilającego. 
Moduł LITEcomp komunikuje się 
z układem DS1807 poprzez magi-
stralę I2C zrealizowaną na wypro-
wadzeniach PB5 i PB4.

Układ DS1807
Wewnętrzną budowę układu 

DS1807 przedstawiono na rys. 2. 
Układ ma trzy wejścia adresowe 
A2…A0, dzięki czemu na jednej 
magistrali I2C można połączyć aż 8 
układów DS1807, co da w efekcie 
możliwość regulacji tłumienia w 16 
niezależnych kanałach. Wewnątrz 
układu DS1807 znajdują się dwa 
65–pozycyjne potencjometry o cha-
rakterystyce logarytmicznej sterowa-
ne magistralą I2C. Budowę rejestru 

pozycji suwaka przedstawiono na 
rys. 3. Regulacja tłumienia odbywa 

Rys. 1.

Rys. 2.

Tab. 1. Zestawienie poleceń obsługiwanych przez sterownik układu DS1807
Rozkaz Kod rozkazu Opis

Write Potentiometer–0 1010 1001 Zapis pozycji suwaka potencjometru 0
Write Potentiometer–1 1010 1010 Zapis pozycji suwaka potencjometru 1

Write Both Pots 1010 1111 Zapis pozycji suwaków obydwu potencjometrów

się z krokiem 1 dB w zakresie 0…
–63 dB (bity 5…0). Możliwe jest 

również uzyskanie tłumienia ponad 
–90 dB poprzez ustawienie bitu 
MUTE (bit 6) w bajcie, przy czym 
wartość pozostałych bitów nie ma 
wpływu na wartość tłumienia. Bit 
7 nie ma wpływu na wartość tłu-
mienia sygnału. Do modyfikacji sta-
nu rejestrów pozycji suwaka służą 
rozkazy przedstawione w tab. 1.

Przebiegi na magistrali I2C pod-
czas zapisu każdego z tych roz-
kazów przedstawiono na rys. 4. 
Istnieje możliwość odczytania ak-
tualnego stanu rejestru konfigura-
cyjnego każdego z potencjometrów. 
Przebiegi na magistrali podczas od-
czytu przedstawiono na rys. 5.

Procedury zaczerpnięte 
z poprzednich projektów

Procedury obsługi wyświetlacza 
LCD omówiliśmy w EP5/2007, zosta-
ły zastosowane bez większych mo-
dyfikacji. Jedyną istotną modyfikacją 

Rys. 3.

Rys. 4.

Rys. 5.

List. 1. Procedura zapisu wartości tłumienia do układu DS1807
;***************************************************************************
; Procedura zapisu wartości tłumienia do układu DS1807
;***************************************************************************
setVolume
 CALL i2c_start    ; start
 LD A, #$50      ; adres układu DS1807
 CALL i2c_write
 LD A, #$AF      ; polecenie WRITE BOTH POTS
 CALL i2c_write
 LD A, volume     ; zapis wartości tłumienia
 CALL i2c_write
 CALL i2c_stop    ; stop
 RET

List. 2. Procedura wyświetlenia wartości tłumienia na wyświetlaczu LCD
;***************************************************************************
; Procedura wyświetlenia wartości tłumienia na wyświetlaczu LCD
;***************************************************************************
displayVolume
 LD A, #4
 CALL lcdsetaddress   ; ustawienie współrzędnych
 WTXT tekstminus
 LD A, volume
 CALL writedecimal   ; wyświetlenie wartości tłumienia
 WTXT tekst1     ; wyświetlenie tekstu „dB”
 RET

List. 3. Procedura zwiększająca tłumienie (zmniejszająca głośność)
;***************************************************************************
; Procedura zwiększająca tłumienie (zmniejszająca głośność)
;***************************************************************************
volDown
 LD A, volume     ;
 CP A, #63      ; czy osiągnięto maksymalnie tłumienie?
 JREQ downend     ; jeśli tak to skok do downend  
 ADD A, #1      ; w przeciwnym razie zwiększamy tłumienie
 LD volume, A     ;
 CALL setVolume    ; i zapisujemy do układu DS1807
 CALL displayVolume
downend
 RET

jest zmiana nazwy makra 
wyświetlającego tekst na 
wyświetlaczu z writetext na 
WTXT. Procedury komuni-
kacji poprzez magistralę 
I2C zostały przedstawione 

w EP6/2007 przy okazji omawiania 
zegara z układem PCF8583. Proce-
dury te zostały zastosowane w bie-
żącym przykładzie bez żadnych 
zmian, poza zmienioną konfiguracją 
wyprowadzeń SDA i SCL. Odbiór 
sygnału zdalnego sterowania w ko-
dzie RC5 przedstawiony był w EP9/
2007, niezbędne procedury zastoso-
wano w tym projekcie bez żadnych 
modyfikacji. Jedyną nowością przed-
stawianego projektu są procedury 
odpowiedzialne za zapis do ukła-
du DS1807 odpowiednich wartości 
określających położenie suwaka po-
tencjometru, które zostaną omówio-
ne w dalszej części artykułu.

Zapis wartości tłumienia
Jedną z najważniejszych proce-

dur jest procedura zapisująca do 
układu DS1807 wartość tłumie-
nia sygnału, czyli pozycję „su-
waka” cyfrowego potencjometru. 
Wartość tłumienia jest zapisywana 
jednocześnie do obu potencjome-
trów przy użyciu polecenia WRITE 
BOTH POTS. Kod procedury przed-
stawiono na list. 1.

Wyświetlenie wartości 
tłumienia

Po zapisaniu nowej wartości 
tłumienia do układu DS1807 na-
stępuje jej wyświetlenie na wy-
świetlaczu LCD modułu LITEcomp. 
Do wyświetlenia liczbowej wartości 
tłumienia wykorzystana jest proce-
dura writedecimal, która została 
omówiona w EP5/2007. Kod proce-
dury displayVolume przedstawiono 
na list. 2.

List. 4. Procedura zmniejszająca tłumienie (zwiększająca głośność)
;***************************************************************************
; Procedura zmniejszająca tłumienie (zwiększająca głośność)
;***************************************************************************
volUp
 LD A, volume     ; 
 CP A, #0       ; czy osiągnięto minimalne tłumienie?
 JREQ upend      ; jeśli tak to skok do upend
 SUB A, #1      ; w przeciwnym razie zmniejszamy tłumienie
 LD volume, A   
 CALL setVolume    ; i zapisujemy do układu DS1807
 CALL displayVolume   ; oraz wyświetlamy na wyświetlaczu LCD
upend  
 RET

List. 8. Program główny zastosowany w prezentowanym projekcie
;**************************************************************************
; Początek programu głównego
;**************************************************************************
._reset
 RSP
 LD PBOR, A
 CALL RC5Init    ; inicjalizacja peryferiów odpowiedzialnych za RC5
 CALL lcdinit    ; inicjalizacja wyświetlacza
 LD A, #64
 CALL lcdsetaddress  ; ustawienie współrzędnych
 WTXT tekst2    ; wyświetlenie adresy strony EP
    
 LD A, #16     ; 
 LD volume, A
 CALL setVolume   ; zapis wartości tłumienia do DS1807
 CALL displayVolume  ; wyświetlenie wartości tłumienia na LCD
 BSET EICR, #7
 BSET PBOR, #0
 RIM       ; odblokowanie przerwań

mainLoop
 CALL WaitForRC5   ; oczekiwanie na odebranie kodu RC5
 LD A, RC5Command
 CP A, #16     ; czy odebrano rozkaz 16 (Volume Up)?
 JRNE dalej1    ; jeśli nie to skok do dalej1
 CALL volUp     ; w przeciwnym razie zmniejszenie tłumienia
dalej
 LD A, RC5Command 
 CP A, #17     ; czy odebrano rozkaz 17 (Volume Down)?
 JRNE dalej2    ; jeśli nie to skok do dalej2
 CALL volDown    ; w przeciwnym razie zwiększenie tłumienia
dalej2
 LD A, RC5Command
 CP A, #13     ; czy odebrano rozkaz 13 (Mute)?
 JRNE dalej3    ; jeśli nie to skok do dalej3
 CALL mute     ; w przeciwnym razie wyciszenie sygnału
dalej3
 JRA  mainLoop   ; skok na początek pętli głównej

List. 5. Procedura ustawiająca maksymalne tłumienie 
(mute)
;*********************************************************
; Procedura ustawiająca maksymalne tłumienie (mute)
;*********************************************************
mute 
 CALL i2c_start    ; start
 LD A, #$51      ; adres układu DS1807 + R
 CALL i2c_write  
 SCF        ; NOACK
 CALL i2c_read    ; odczyt stanu potencjometrów
 LD muteVol, A
 CALL i2c_stop    ; stop
  
 CALL i2c_start    ; start
 LD A, #$50      ; adres układu DS1807
 CALL i2c_write
 LD A, #$AF      ; polecenie WRITE BOTH POTS
 CALL i2c_write
 LD A, muteVol
 XOR A, #%01000000  
 CALL i2c_write
 CALL i2c_stop    ; stop
    
 LD A, muteVol
 AND A, #%01000000   ; czy mute włączone?
 JRNE noMute     ; jeśli nie to skok do noMute
 CALL displayMute   ; w przeciwnym razie
 JRA mute1
noMute
 CALL displayVolume
mute1
 RET

List. 6. Procedura wyświetlająca informację o wycisze-
niu na wyświetlaczu LCD
;*****************************************************
; Procedura wyświetlająca informację o wyciszeniu na 
wyświetlaczu LCD
;*****************************************************
displayMute
 LD A, #4
 CALL lcdsetaddress   ; ustawienie współrzędnych
 WTXT muteTekst    ; wyświetlenie tekstu
 RET

Zwiększenie wartości 
tłumienia

Odebranie kodu RC5 z rozka-
zem o wartości 17 lub naciśnięcie 
przycisku S2 spowoduje wywołanie 
procedury zwiększającej tłumienie, 
co objawi się spadkiem głośności 
sygnału na wyjściu. Kod procedury 
jest przedstawiony na list. 3.

Na początku procedury nastę-
puje porównanie aktualnej wartości 
tłumienia z liczbą 63 w celu spraw-
dzenia, czy osiągnięto maksymalną 
wartość tłumienia. Jeśli operacja po-
równania wykaże, iż wartość tłumie-
nia ma maksymalną możliwą war-
tość następuje zwiększenie wartości 
zmiennej volume, która odpowiada 
za wartość tłumienia potencjometru.

Zmniejszenie wartości 
tłumienia

Odebranie kodu RC z rozkazem 
o wartości 16 lub naciśnięcie przy-
cisku S1 spowoduje wywołanie 
procedury zmniejszającej tłumienie 
(list. 4).

Na początku procedury wykony-
wane jest sprawdzenie, czy osiąg-
nięto minimalną wartość tłumienia. 
Jeśli nie to następuje zwiększenie 
wartości zmiennej volume, w prze-
ciwnym razie następuje opuszcze-
nie procedury.

Wyciszenie sygnału (mute)
Czasami zachodzi potrzeba na-

tychmiastowego wyciszenia sygnału 
audio, które jest realizowane funkcją 

List. 7. Procedura obsługi przerwania EI3 od przycisków 
S1 i S2
;*****************************************************
; Procedura obsługi przerwania EI3 od przycisków S1 
i S2
;*****************************************************
._EI3Vect
 BTJT PADR, #0, EI3dalej1
 CALL volUp
 IRET
EI3dalej1
 BTJT PADR, #2, EI3dalej2
 CALL volDown
EI3dalej2
 IRET

mute. Układ DS1807 umożliwia cał-
kowite wyciszenie sygnału poprzez 
ustawienie bitu 6 w rejestrze odpo-
wiedzialnym za położenie wirtualne-
go suwaka potencjometru. Każdora-
zowe odebranie przez program kodu 
RC5 o rozkazie 13 spowoduje zane-
gowanie stanu tego bitu, uprzednio 
odczytanego z układu DS1807. Do-
datkowo na wyświetlaczu LCD mo-

dułu LITEcomp zostanie wyświetlo-
na informacja o włączeniu funkcji 
mute. Wyłączenie funkcji mute jest 
możliwe poprzez kolejne odebranie 
kodu 13, albo naciśnięcie przyci-
sków regulacji głośności, zarówno 
na klawiaturze lokalnej modułu, jak 
i na pilocie zdalnego sterowania. 
Kod procedury mute przedstawiono 
na list. 5.

Procedura mute wywołuje w za-
leżności od stanu bitu MUTE pro-
cedurę displayVolume, wyświetla-
jącą aktualne tłumienie sygnału, 
albo procedurę displayMute, wy-
świetlającą tekst MUTE ON. Kod 
procedury displayMute przedstawio-
no na list. 6.

Procedura obsługi przerwania 
zewnętrznego EI3

Przyciski S1 i S2 zostały wy-
korzystane do regulacji tłumienia. 
Naciśnięcie przycisku S1 spowo-
duje zmniejszenie tłumienia, na-
tomiast naciśnięcie przycisku S2 
spowoduje zwiększenie tłumienia. 
Ponieważ program w pętli głównej 
oczekuje na ustawienie flagi ode-
brania ramki sygnału RC5, obsłu-
ga przycisków została zrealizowana 
w przerwaniu. Przyciski S1 i S2 
oprócz podłączenia odpowiednio 
do wyprowadzeń PA0 i PA2 zostały 
również podłączone poprzez bram-
kę AND zrealizowaną na diodach 
D1 i D2 oraz na rezystorze R5   
do wyprowadzenia PB0, należącego 
do grupy przerwania zewnętrzne-

R E K L A M A

go EI3. Naciśnięcie któregokolwiek 
z przycisków spowoduje wymusze-
nie na wyprowadzeniu PB0 zbocza 
opadającego, i w jego efekcie wystą-
pi przerwanie. W ramach procedu-
ry obsługi tego przerwania zostanie 
sprawdzony stan przycisków S1 
i S2 i w zależności od tego, który 
z nich został przyciśnięty, nastąpi 
wywołanie albo procedury zmniej-
szającej tłumienie, albo procedury 
zwiększającej tłumienie. Kod proce-
dury obsługi przerwania EI3 przed-
stawiono na list. 7.

Program główny
Główna część programu rozpo-

czyna się od inicjalizacji wyświe-
tlacza LCD oraz układów peryferyj-
nych wykorzystywanych do odbio-
ru kodu RC5. Następnie w drugiej 
linii wyświetlacza LCD wyświetlany 
jest stały tekst (adres strony inter-
netowej EP), a następnie ustawiane 
jest tłumienie na poziomie –16 dB. 
W pętli głównej program oczekuje 
na odebranie kodu RC5, a następ-
nie jest sprawdzany odebrany kod. 
W obecnej postaci programu nie 

jest sprawdzany odebrany adres, 
tak więc poprawny będzie każdy 
odebrany kod, którego rozkaz ma 
wartość 13 (Mute), 16 (Volume Up) 
lub 17 (Volume Down), bez wzglę-
du na odebrany adres urządzenia. 
Umożliwia to sterowanie regulato-
rem za pomocą pilota od dowolne-
go sprzętu, np. telewizora, czy też 
sprzętu audio. Kod programu głów-
nego przedstawiono na list. 8.

Przedstawiony w artykule pro-
gram wraz z układem sprzęto-
wym stanowi w pełni funkcjonalny 
układ regulatora głośności sygnału 
audio. Jednakże w obecnych cza-
sach sama regulacja głośności nie 
jest już czymś szczególnym. Trady-
cyjnie więc zachęcam Czytelników 
do samodzielnych eksperymentów 
oraz rozwijania programu jak i po-
szerzania funkcjonalności układu 
o dodatkowe funkcje spełniające 
wymagania współczesnego sprzętu 
audio.
Radosław Kwiecień, EP
radoslaw.kwiecien@ep.com.pl


